
ABSTRACT 

When dealing with data from multiple or unstructured data sources, there can be data set variable attribute conflicts. 
These conflicts can be cumbersome to deal with when developing code. This paper intends to discuss issues and 
table driven strategies for dealing with data sets with inconsistent variable attributes. 
 

INTRODUCTION 

This paper intends to provide information to SAS® programmers of all levels and help them resolve issues when 
combining data that has content that is appropriate for combining but variable attributes that are not identical. 
Depending on the situation with your data and the method you are using to combine the data, inconsistent variable 
attributes can significantly affect the results from your program. This paper will also show you how to set up a data 
set with the desired attributes and then use it as the standard template for others programmatically. 

One disclaimer for this paper is that at the time this paper was written, I only had brief exposure to SAS 9.4. There 
could be features and functionality that I am not familiar with. 

DATA SET AND VARIABLE ATTRIBUTES 

SAS data set and variable attributes can be viewed by examining results from a PROC CONTENTS. Consider the 
program and results shown below. The PROC CONTENTS results are show in 4 figures based on the how the output 
is organized. For the purpose of this paper, I have used SASHELP.ZIPCODE. It should be readily available within 
any SAS environment. 

 

proc contents data=sashelp.zipcode; 

run; 

 

 

 

Figure 1. Top Section of SAS Data Set Attributes from PROC CONTENTS Output  

Figure 2. Second Section of SAS Data Set Attributes from PROC CONTENTS Output 

 

 

Figure 3. Section of Variable Attributes from PROC CONTENTS Output 

Figure 4. Bottom Section with Additional Data Set Attributes from PROC CONTENTS Output 

 

SAS DATA SET ATTRIBUTES 

Referencing the output shown in the four previous figures, the SAS data set attributes are shown in figures 1, 2, and 
4. They are not the focus of this paper but they have other issues depending on how you are working with SAS Data 
Sets. 

One important issue with SAS Data Set attributes is that they do not persist if you re-create a SAS Data Set during 
processing. Consider the code below… 

data zipcodes; 

   set zipcodes; 

run; 

 

Or 

 

proc sql; 

   create table zipcodes as 

   select * 

   from zipcodes; 

quit; 

 

If you were to run the above code, any data set attributes that you have added such as data set labels, indexes, 
constraints, flags that indicate the data set has been sorted, etc. will be lost.  

 

SAS VARIABLE ATTRIBUTES 

Figure 3 from shows the variable attributes from the SAS Data Set SASHELP.ZIPCODES that are displayed by 
PROC CONTENTS. This paper will be focusing on the variable attributes type, length, and format but to be complete 
I have leveraged SASHELP.VCOLUMN to show all of the variable attributes contained in the SAS Dictionary Tables.  

proc print data=sashelp.vcolumn; 

where libname=’SASHELP’ and memname=’ZIPCODE’; 

run; 

 

This information can also be leveraged from DICTIONARY.COLUMNS if you are using PROC SQL. 

proc sql; 

   select * 

   from dictionary.columns 

   where libname=’SASHELP’ and memname=’ZIPCODE’; 

quit; 

 

The data set SASHELP.VCOLUMN is actually a view that points at DICTIONARY.COLUMNS. Additionally, 
SASHELP.VTABLE and DICTIONARY.TABLES contain information about the SAS data set attributes. There are 
many other dictionary tables. You can see the list of all dictionary tables by querying DICTIONARY.DICTIONARIES. 

To learn more about the SAS Dictionary Tables, please reference the SAS web site.  

http://support.sas.com/documentation/cdl/en/lrcon/64801/HTML/default/viewer.htm#p11h9yhja6t25an1mkx8xgy3wcw
m.htm 

 

 

Figure 5. Listing from SASHELP.VCOLUMN for the SAS Data Set SASHELP.ZIPCODE 

 

Now that we have looked at all of the SAS data set and variable attributes, let us re-focus to the variable attributes of 
interest for this paper (type, length, and format). When combining SAS Data Sets with inconsistent type, length, and 
format definitions, results can be compromised. Next we will discuss the methods for combining SAS Data Sets that 
are relevant to this paper. 

METHODS FOR COMBINING SAS DATA SETS 

At this point it is appropriate to discuss some of the different methods for combining SAS data sets. To start off with, 
let’s discuss the three different scenarios for combining data; adding rows, adding columns, and modifying existing 
values. Please note that there are probably more methods available than discussed in this paper. We will mention 
three different data combination scenarios but in this paper we will focus on the Adding Rows scenario. 

 

Modifying Existing Data Values 

This type of operation is typically a master transaction type of update. You have two data sets. The first data set has 
all of the data and the other has some records with values that are meant to update a few values in the first data set 
by using primary keys to indicate which records are updated. 

 

Adding Columns 

This type of operation is typically a join/merge operation. You have two or more data sets with different variables 
except for the keys that are appropriate for matching up the records. In SAS terms this is known as a Data Step 
Merge or a PROC SQL Join. 

 

Adding Rows 

This type of operation is typically a stacking operation. You have multiple data sets with the same variables and you 
want to stack them. In SAS terms this is known as a Data Step SET, PROC APPEND, or PROC SQL OUTER UNION 
CORRESPONDING. 

COMBINING SAS DATA SETS WITH NON-SAS DATA 

Although the focus of this paper is not how to read in data from non-SAS data sources, it is worth mentioning the 
potential issues with some of the non-SAS sources. 

 

EXCEL 

Although an Excel worksheet looks and feels like a table, the attributes for a column do not necessarily need to be 
consistent. Attributes can change on a cell to cell basis thus making data extracts somewhat unpredictable. By 
default, SAS uses the first eight rows of data in a column to determine the optimal variable attributes. 

 

ASCII / EBCDIC 

Although ASCII / EBCDIC data does not have built in data set or variable attributes, you have complete control over 
how the data is read in. By using INFILE and INPUT statements, you can control the name, type, length and format of 
each variable. 

 

OTHER STRUCTURED DBMS 

This is the best non-SAS data to read in as structured databases enforce column attributes like in SAS data sets. 

 

CREATING A TEMPLATE DATA SET 

A useful approach in combining data sets with like variables but potentially inconsistent attributes is to create a SAS 
data set that has the desired data set and variable attributes but without any actual data records. There are a few 
methods for achieving this in SAS. 
 
 

MANUAL METHODS 

Using the SAS Data Step, you can create an empty SAS Data Set with the exact desired variable attributes. 

 

Data Step Method 

data class; 

   attrib name   length=$8 label=’Name’ 

          sex    length=$1 label=’Sex’ 

          age    length=8  label=’Age’        format=3. 

          height length=8  label=’Height CM’  format=5.1 

          weight length=8  label=’Weight lbs’ format=5.1 

   ; 

   stop; 

run; 

 
 
PROC SQL Method 
 

proc sql; 

   create table class 

   (name   char(8) label=’Name’, 

    sex    char(1) label=’Sex’, 

    age    num     label=’Age’        format=3., 

    height num     label=’Height CM’  format=5.1, 

    weight num     label=’Weight lbs’ format=5.1 

    ); 

quit; 

AUTOMATICALLY BY USING ANOTHER SAS DATA SET 

If there is another SAS data set which has the desired attributes, it is not efficient to re-invent the wheel as we did in 
the previous section. It is more efficient to point at another SAS Data Set that has the variable attributes that we 
desire. 

 

Data Step Method 

data zipcode; 

   if 0 then 

      set sashelp.zipcode; 

   stop; 

run; 

 
 
PROC SQL Method 
 

proc sql; 

   create table class like sashelp.zipcode; 

quit; 

 

HANDLING ATTRIBUTE INCONSISTENCIES WHILE APPENDING 

Since the focus of this paper is stacking SAS data sets with the same variables, this section will only focus on Data 
Step Set, PROC APPEND, and PROC SQL OUTER UNION CORRESPONDING. 

 

VARIABLE TYPE MISMATCHES 

Variable type mismatches will stop a data set stacking process in its tracks. A variable type mismatch happens when 
one data set has a variable that is type numeric and then another data set has a variable that is type character. A 
program will encounter an error condition and the data will need to be fixed. This is true regardless of the method 
used to combine the SAS Data Sets. 
 

VARIABLE LENGTH, FORMAT, AND LABEL MISMATCHES 

It is possible to get around variable length, format, and label mismatches; but you probably want them to be 
consistent to avoid data quality and/or value truncation problems. As long as the variable types are the same and the 
variables have the same name, PROC SQL OUTER UNION CORRESPONDING will use the longest length across 
all of the data sets while Data Step and PROC APPEND use the first reference to the variable based on the order of 
the data sets. 
 

WHAT TO DO? 

Ask for the source data with the appropriate attribute so we do not have this problem in the first place. Often we do 
not have control over this and the entity sending the data will not conform. We could spend significant time working 
with each data source manually and get the attributes consistent. Or we could use our data drive skills along with the 
SAS Metadata contained in the SAS Dictionary Tables.

A DATA DRIVEN STRATEGY 
It is possible to get around variable type, length, format, and label mismatches programmatically by using the SAS 
metadata. One method uses the SAS Dictionary Tables. The full code will be shared in the presentation but a sample 
is below. This sample code assumes that the SAS Data Set SASHELP.CLASS has the desired SAS variable 
attributes and we create NEWCLASSDATA as a hypothetical data set that we want to append to it. 
 

*************************************************************************; 

** Create a hypothetical data set with different attributes 

*************************************************************************; 

data newclassdata; 

   length name $20 sex $8 height 4 weight 4; 

   label name='Student Name'; 

   format height words60.; 

   set sashelp.class(rename=(age=tage)); 

   age=left(put(tage,2.)); 

run; 

 

*************************************************************************; 

** Read the SAS Data Set Metadata into a SAS Data Set so we can work 

** with it programmatically and in a data driven manner. 

*************************************************************************; 

proc sql; 

   create table compare_attributes as 

   select target.name, 

          target.type   as target_type, 

          new.type      as new_type, 

          target.length as target_length, 

          new.length    as new_length, 

          target.format as target_format, 

          new.format    as new_format, 

          target.label  as target_label, 

          new.label     as new_label 

   from (select *  

         from dictionary.columns   

         where libname='SASHELP' and memname='CLASS')    as target, 

        (select *  

         from dictionary.columns      

         where libname='WORK' and memname='NEWCLASSDATA') as new 

   where lowcase(target.name)=lowcase(new.name); 

quit; 

   

Now that the SAS variable attributes are in a SAS Data Set, we can make decisions and generate code in a data 
driven manner. 
 

 
Figure 6. COMPARE_ATTRIBUTES Data Set Resulting from previous PROC SQL. 
 
We can generate code to … 

• recognize type mismatches and generate an input or put function 

• recognize the maximum length and generate a length statement to ensure data truncation will not take place 

• use the target format or generate an appropriate format based on the desired length 

• use the target label 

CONCLUSION 

Using the information and methods in this paper should result in a more stable process for combining SAS data sets. 
This will improve the consistency and quality of the data you want to use for analysis. 

